應變計試驗配合最小二乘法計算複合材料之材料係數

朱聖浩*，**李官峰****
國立成功大學土木工程學系
***教授，** 研究生**

摘要

本論文使用最小二乘法來計算複合材料之應變及應力關係，本法之優點為方法簡易且具系統化，適合使用電腦，重複輸入施加之單軸張力，兩個應變計量到之應變及兩個與試體相關之角度，即可計算出材料之應變及應力關係矩陣。由理論配合數值之驗證得知：本法之平均計算誤差約為輸入之應變誤差上限之一半，若輸入之應變誤差不大，本法可計算出精確之材料係數。

1.前言

複合材料為一綜合性產業，優點為加工種類多、設計自由度、單位質量輕、耐蝕性佳，屬功能性、結構性材料，在現今社會使用極為廣泛。複合材料之構成可粗分為基材(Matrix)和補強(Rein-forcement)，其中以纖維(fiber)補強之複合材料最為常見，例如碳纖維及玻璃纖維。複合材料為非均向性材料，其應變及應力之關係比均向材料複雜，要了解複合材料首先就是知道複合材料之應變及應力關係，Rosen (1972)將試體沿纖維之 45° 方向切割，在用單軸張力求得複合材料之剪力模數。Christos 及 Sinclair (1977)將試體沿纖維之 10° 方向切割做了複合材料實驗，探討施加軸向拉力。藉由應力元素角度轉換公式和應變角度轉換公式，求取剪力模數及楊氏係數，這些實驗方式已為各界所使用，去求取複合材料之應變及應力關係(Kobayashi, 1993)。本論文則使用最小二乘法來計算複合材料之應變及應力關係，本方法簡易且具系統化，適合使用電腦，重複輸入施加之單軸張力，兩個應變計量測到之應變及兩個與試體相關之角度，即可計算出材料之應變及應力關係矩陣，而其中部份之隨機誤差亦會相互抵消。

2. 最小二乘法計算複合材料係數



圖 1.複合材料試驗試體之分割



圖 2. 試驗試體之座標說明

使用應變計求取複合材料之應變及應力關係矩陣，一般至少需使用三個試體，分別在一塊大試片上從三個方向切割下來．圖 1.**顯示試驗試體之可能分割方式，其中** 1-2 **軸為大域座標，欲求之應變及應力關係矩陣即以此兩軸為基準，兩個分割試體一般為平行** 1 **及** 2 **軸切割，第三個試體則無法平行** 1 **或** 2 **軸。圖** 2 **顯示某個試體，其中** 1-2 **軸為大域座標，與圖** 1 **中之** 1-2 **座標相同，**x-y **軸為試體方向，**1 **軸至** x **軸之夾角為**θ(**逆時針為正**)**，而均勻外加張應力**(σ_0)**作用在** y **軸上，**a-b **軸為兩個應變計之方向，**x **軸至** a **軸之夾角為**α (**逆時針為正**)**。式**(1)**顯示以** 1-2 **軸為基準之平面上應變及應力關係，**

$$\{\varepsilon_{12}\} = [C_{12}]\{\sigma_{12}\} \qquad (1)$$

其中$\{\varepsilon_{12}\}$ **為** 1-2 **軸之應變向量，**$\{\sigma_{12}\}$ **為** 1-2 **軸之應力向量，及**$[C_{12}]$ **以** 1-2 **軸為基準之平面上應變及應力關係矩陣。**

由於 x-y **軸之應力及** a-b **軸之應變為已知之數據，故執行下兩式之座標轉換，**

$$\{\varepsilon_{12}\} = [T_{\varepsilon a}]^{-1}\{\varepsilon_{ab}\} \qquad (2)$$

$$\{\sigma_{12}\} = [T_{\varepsilon}]^{T}\{\sigma_{xy}\} \qquad (3)$$

其中
$$[T_{\varepsilon a}] = \begin{bmatrix} \cos^2(\alpha+\theta) & \sin^2(\alpha+\theta) & \cos(\alpha+\theta)\sin(\alpha+\theta) \\ \sin^2(\alpha+\theta) & \cos^2(\alpha+\theta) & -\cos(\alpha+\theta)\sin(\alpha+\theta) \\ -2\cos(\alpha+\theta)\sin(\alpha+\theta) & 2\cos(\alpha+\theta)\sin(\alpha+\theta) & \cos^2(\alpha+\theta)-\sin^2(\alpha+\theta) \end{bmatrix} \qquad (4)$$

$$[T_{\varepsilon}]^{T} = \begin{bmatrix} \cos^2(\theta) & \sin^2(\theta) & -2\cos(\theta)\sin(\theta) \\ \sin^2(\theta) & \cos^2(\theta) & 2\cos(\theta)\sin(\theta) \\ \cos(\theta)\sin(\theta) & -\cos(\theta)\sin(\theta) & \cos^2(\theta)-\sin^2(\theta) \end{bmatrix} \qquad (5)$$

將(2)**和**(3)**式代入**(1)**式，得到**

$$[T_{\varepsilon a}]^{-1}\{\varepsilon_{ab}\} = [C_{12}]\{T_{\varepsilon}\}^{T}\{\sigma_{xy}\} \qquad (6)$$

再將上式兩邊乘上 $[T_{\varepsilon a}]$**，則**

$$\{\varepsilon_{ab}\} = [T_{\varepsilon a}][C_{12}][T_{\varepsilon}]^{T}\{\sigma_{xy}\} \qquad (7)$$

假設
$$\begin{Bmatrix} r \\ s \\ t \end{Bmatrix} = [T_{\varepsilon}]^{T}\{\sigma_{xy}\} = [T_{\varepsilon}]^{T}\begin{Bmatrix} 0 \\ \sigma_0 \\ 0 \end{Bmatrix}, \qquad (8)$$

$$[T_{\varepsilon a}] = \begin{bmatrix} T_{11} & T_{12} & T_{13} \\ T_{21} & T_{22} & T_{23} \\ T_{31} & T_{32} & T_{33} \end{bmatrix} \qquad (9)$$

及
$$[C_{12}] = \begin{bmatrix} Q_{11} & Q_{12} & Q_{16} \\ Q_{12} & Q_{22} & Q_{26} \\ Q_{16} & Q_{26} & Q_{66} \end{bmatrix} \qquad (10)$$

將式(8)**至**(10)**代入**(7)**得**

$$\begin{Bmatrix}\varepsilon_a\\ \varepsilon_b\\ \gamma_{ab}\end{Bmatrix}=\begin{bmatrix}T_{11} & T_{12} & T_{13}\\ T_{21} & T_{22} & T_{23}\\ T_{31} & T_{32} & T_{33}\end{bmatrix}\begin{bmatrix}Q_{11} & Q_{12} & Q_{16}\\ Q_{12} & Q_{22} & Q_{26}\\ Q_{16} & Q_{26} & Q_{66}\end{bmatrix}\begin{Bmatrix}r\\ s\\ t\end{Bmatrix} \tag{11}$$

再將$[C_{12}]$和$\{\sigma_{12}\}$相乘，得到

$$\begin{Bmatrix}\varepsilon_a\\ \varepsilon_b\\ \gamma_{ab}\end{Bmatrix}=\begin{bmatrix}T_{11} & T_{12} & T_{13}\\ T_{21} & T_{22} & T_{23}\\ T_{31} & T_{32} & T_{33}\end{bmatrix}\begin{bmatrix}rQ_{11}+sQ_{12}+tQ_{16}\\ rQ_{12}+sQ_{22}+tQ_{26}\\ rQ_{16}+sQ_{26}+tQ_{66}\end{bmatrix} \tag{12}$$

由於使用應變計只量測ε_a及ε_b，上式再整理如下：

$$\begin{Bmatrix}\varepsilon_a\\ \varepsilon_b\end{Bmatrix}==\begin{bmatrix}T_{11}rQ_{11}+T_{11}sQ_{12}+T_{11}tQ_{16}+T_{12}rQ_{12}+T_{12}sQ_{22}+T_{12}tQ_{26}+T_{13}rQ_{16}+T_{13}sQ_{26}+T_{13}tQ_{66}\\ T_{21}rQ_{11}+T_{21}sQ_{12}+T_{21}tQ_{16}+T_{22}rQ_{12}+T_{22}sQ_{22}+T_{22}tQ_{26}+T_{23}rQ_{16}+T_{23}sQ_{26}+T_{23}tQ_{66}\end{bmatrix} \tag{13}$$

將複合材料係數矩陣提出成 6×1 矩陣，如下所示：

$$\begin{Bmatrix}\varepsilon_a\\ \varepsilon_b\end{Bmatrix}=\begin{bmatrix}T_{11}r & T_{11}s+T_{12}r & T_{12}s & T_{11}t+T_{13}r & T_{12}t+T_{13}s & T_{13}t\\ T_{21}r & T_{21}s+T_{22}r & T_{22}s & T_{21}t+T_{23}r & T_{22}t+T_{23}s & T_{23}t\end{bmatrix}\begin{Bmatrix}Q_{11}\\ Q_{12}\\ Q_{22}\\ Q_{16}\\ Q_{26}\\ Q_{66}\end{Bmatrix} \tag{14}$$

令$\{\varepsilon_{ab}\}=\begin{Bmatrix}\varepsilon_a\\ \varepsilon_b\end{Bmatrix}$**，**$[A]=\begin{bmatrix}T_{11}r & T_{11}s+T_{12}r & T_{12}s & T_{11}t+T_{13}r & T_{12}t+T_{13}s & T_{13}t\\ T_{21}r & T_{21}s+T_{22}r & T_{22}s & T_{21}t+T_{23}r & T_{22}t+T_{23}s & T_{23}t\end{bmatrix}$**及**$\{Q\}=\begin{Bmatrix}Q_{11}\\ Q_{12}\\ Q_{22}\\ Q_{16}\\ Q_{26}\\ Q_{66}\end{Bmatrix}$**。**

得到$\{\varepsilon_{ab}\}=[A]\{Q\}$ (15)

每個試體在某個施加張力(σ_0)下(此處定義為第 i 次結果)，均可得到(15)式之結果，接下來可使用最小二乘法來求取$\{Q\}$。如果共有 N 次結果，則殘餘誤差可定義為：

$$\pi=\sum_{i=1}^{N}\left([A]_i\{Q\}-\{\varepsilon_{ab}\}_i\right)^2 \tag{16}$$

將上式對{Q}微分得：

$$\left(\sum_{i=1}^{N}[A]_i^T[A]_i\right)\{Q\}=\sum_{i}^{N}[A]_i^T\{\varepsilon_{ab}\}_i \tag{17}$$

(17)式為一線性方程式，求解後得$\{Q\}$。若$\{Q\}$之某幾項已知為零，例如對稱纖維安排之複合材料，其 $Q_{16}=Q_{26}=0$，此時在(14)式中將這些項移除，而不必在最小二乘法中計算，可增加計算之精度。

3.使用理論解驗證最小二乘法之精確度

本節使用理論解驗證最小二乘法之精確度，三種材料係數設定如表 1。每種材料切割成三個試體，其角度分別為：試體 1($\theta=0^o$, $\alpha=0^o$)，試體 2($\theta=90^o$, $\alpha=30^o$)，及試體 3($\theta=45^o$, $\alpha=0^o$)，外力(σ_0)使用 0.001, 0.002, 0.005 及 0.01GPa。由假設之施加應力(σ_0)，角度θ及α，與由理論計算出之應變ε_a 及ε_b，

經使用(17)式或附錄上之程式，得到幾乎無誤差之材料係數{Q}。若將應變ε_a及ε_b加上人為誤差，即

$$x_{new} = (1 + c_x)x_{old} \tag{18}$$

其中 x_{old} 為輸入之應變ε_a及ε_b，x_{new} 為含誤差之應變ε_a及ε_b，及 c_x 為一隨機數介於 -c 至 c 之間(c 是誤差上限，為一正實數)。為執行誤差探討，茲定義最小二乘法之計算誤差(error)為

$$\text{error} = \left(\sum \left| Q_{ij}^{LS} - Q_{ij} \right|\right) / \left(\sum \left| Q_{ij} \right|\right) \tag{19}$$

其中 Q_{ij}^{LS} 為最小二乘法之計算出之 6 個材料係數，及 Q_{ij} 為設定之 6 個材料係數(表 1)。使用式(18)含誤差之應變計算材料係數，再與表 1 之正確資料一同代入(19) 式，可繪出圖 3 之結果，此圖為最小二乘法之計算誤差(error，19 式)與誤差上限(c)之關係。本圖說明:

(1)最小二乘法之最大計算誤差約與輸入之應變誤差上限(c)相當。
(2)最小二乘法之平均計算誤差約為輸入之應變誤差上限(c)之一半(0.455 倍)。
(3)若輸入之應變誤差不大，最小二乘法可計算出精確之材料係數。

表 1. 設定之三種材料係數(1/GPa)

材料	Q_{11}	Q_{12}	Q_{16}	Q_{22}	Q_{26}	Q_{66}
1	0.10000E+00	-0.20000E-02	0.00000E+00	0.50000E-02	0.00000E+00	0.20000E+00
2	0.93312E-01	-0.19063E-01	0.21434E-01	0.45813E-01	0.60838E-01	0.13175E+00
3	0.75250E-01	-0.24750E-01	0.47500E-01	0.75250E-01	0.47500E-01	0.10900E+00



圖 3.最小二乘法之計算誤差(error，19 式)與誤差上限(c)之關係(圖中之直線為線性迴歸)


圖 4. 試驗之架設

4. 應變計試驗之計算實例

本節說明如何使用最小二乘法配合實際之試驗，求得複合材料之應變及應力關係矩陣。加載拉力之儀器為 Instron 8801 萬能試驗機，配合應變測定器測得試體應變值之後，再帶入附錄之 Fortran 程式中求取材料係數{Q}。複合材料試體為碳纖維貼片組成(0/45/90/45/0)，試體之分割如圖 1 所示，試體 1(θ=0^o, α=0^o)，試體 2(θ=90^o, α=30^o)，及試體 3(θ=45^o, α=0^o)，試體之長度，寬度及厚度分別為 300，30 及 0.86 公釐，有兩組共六個試體，圖 4 顯示試驗之架設，表

2 顯示施加之應力及量測之應變，表 3 顯示計算之結果。

5. 結論

複合材料之剪力係數與軸力係數不相偶合時，其應變及應力關係較易由試驗計算得到，但若相互偶合時，則不易直接由人為計算得到。本論文使用最小二乘法來計算複合材料之應變及應力關係，本法簡易且具系統化，適合使用電腦，重複輸入施加之單軸張力，兩個應變計量測到之應變及兩個與試體相關之角度，即可計算出材料之應變及應力關係矩陣。由本文驗證得知最小二乘法之平均計算誤差約為輸入之應變誤差上限之一半，若輸入之應變誤差不大，最小二乘法可計算出精確之材料係數。

表 2. 施加之應力(Gpa)及量測之應變($x10^{-6}$)

(a)第一組共三個試體

試體 1($\theta=0^o$, $\alpha=0^o$)			試體 2($\theta=90^o$, $\alpha=30^o$)			試體 3($\theta=45^o$, $\alpha=0^o$)		
σ_0	ε_a	ε_b	σ_0	ε_a	ε_b	σ_0	ε_a	ε_b
0.003951	-5	75	0.009884	149	295	0.010987	-142	230
0.007901	-12	148	0.019769	266	482	0.021973	-270	439
0.011852	-21	221	0.029653	384	670	0.032960	-398	646
0.015802	-27	304	0.039537	498	860	0.043946	-526	856
0.019753	-36	390	0.049422	608	1045	0.054933	-656	1065
0.023703	-46	475	0.059306	718	1237	0.065920	-787	1271
0.027654	-56	558	0.063260	762	1316	0.070314	-841	1357
0.031604	-65	645	0.067213	805	1393	0.074709	-896	1441

(b)第二組共三個試體

試體 1($\theta=0^o$, $\alpha=0^o$)			試體 2($\theta=90^o$, $\alpha=30^o$)			試體 3($\theta=45^o$, $\alpha=0^o$)		
σ_0	ε_a	ε_b	σ_0	ε_a	ε_b	σ_0	ε_a	ε_b
0.004077	-3	60	0.009878	132	223	0.011241	-132	252
0.008154	-6	133	0.019755	265	452	0.022482	-253	472
0.010193	-8	172	0.029633	356	603	0.033723	-373	683
0.012232	-11	206	0.039510	470	795	0.044964	-485	893
0.016309	-18	290	0.049388	587	993	0.056205	-608	1093
0.020386	-27	370	0.059266	722	1225	0.058453	-635	1136
0.024463	-35	454	0.069143	815	1380	0.062950	-690	1225
0.028541	-44	539	0.071119	837	1412	0.067446	-738	1312

表 3. 試驗得到之材料係數(1/GPa)

	Q_{11}	Q_{12}	Q_{16}	Q_{22}	Q_{26}	Q_{66}
第一組試體	0.3531E-01	-0.1946E-02	0.2000E-01	-0.1120E-01	0.4786E-01	0.9935E-01
第二組試體	0.3345E-01	-0.1330E-02	0.1825E-01	-0.1054E-01	0.4221E-01	0.9270E-01
第一及二組試體	0.3440E-01	-0.1679E-02	0.1924E-01	-0.1092E-01	0.4522E-01	0.9621E-01

參考資料

Rosen, B. Walter, 1972. A Simple Procedure for Experimental Determination of the Longitudinal Shear Modulus of Unidirectional Composites, J. Compos. Mater., 6, 552-554.

Chamis, Christos C. and Sinclair, John H., 1977. Ten-Degree Off-Axis Test for Shear Properties in Fiber Composites, Exp. Mech., 17, 339-346.

Kobayashi, Albert S., 1993, Handbook on Experimental Mechanics, 2nd ed., SEM, New York, 858-867.

附錄：Fortran 程式

```
c***********************************************
c**    Program finds composite material constants
c**     The mathod to run the program
c**      MATER inp out
c**     where MATER is the program, inp is the
c**     input data and out is the output result.
c**
        implicit real*8 (a-h,o-z)
        dimension ek(6,6),f(6),A(2,6),e(2)
c**
        ta=datan2(0.0d0,-1.0d0)/180.0d0
        ek=0.0d0
        f=0.0d0
c**
c**     Loop for each specimen result
c**
        do
c**
c**        read test result for each specimen
c**        Sy=applied stress
c**        Ea=strain of axis-a
c**        Eb=strain of axis-b
c**       th=angle (degree) from axis-1 to axis-x
c**       af=angle (degree) from axis-x to axis-a
c**
           read (1,*,iostat=kk) Sy,Ea,Eb,th,af
           if (kk.ne.0) exit
c**
c**        change angles from degree to rad.
c**
           th=th*ta
           af=th+af*ta
c**
c**        find {r,s,t}'=[Te]'{0,Sy,0}'
c**
           cc=cos(th)
           ss=sin(th)
           cs=cc*ss
           cc=cc*cc
           ss=ss*ss
           r=ss*Sy
           s=cc*Sy
           t=-cs*Sy
c**
c**        Find [A] and {E}
c**
           e(1)=Ea
           e(2)=Eb
c**
           cc=cos(af)
           ss=sin(af)
           cs=cc*ss
           cc=cc*cc
           ss=ss*ss
           A(1,1)=cc*r
           A(1,2)=cc*s+ss*r
           A(1,3)=ss*s
           A(1,4)=cc*t+cs*r
           A(1,5)=ss*t+cs*s
           A(1,6)=cs*t
           A(2,1)=ss*r
           A(2,2)=ss*s+cc*r
           A(2,3)=cc*s
           A(2,4)=ss*t-cs*r
           A(2,5)=cc*t-cs*s
           A(2,6)=-cs*t
c**
c**   Find sum([EK]=[A]'[A]) and sum({f}=[A]'{e})
c**
           do i=1,6
              do j=1,6
                 do k=1,2
                  ek(i,j)=ek(i,j)+a(k,i)*a(k,j)
                 enddo
              enddo
              do k=1,2
                 f(i)=f(i)+a(k,i)*e(k)
              enddo
           enddo
        enddo
c**
c**     Solve equation to get material constants
c**
        call solve1(ek,f,6)
c**
       write(2,*)'Q11','Q12','Q22','Q16','Q26','Q66'
        write(2,'(6e12.4)') f
        end
C***********************************************
        SUBROUTINE SOLVE1(A,B,N)
        IMPLICIT REAL*8 (A-H,O-Z)
        DIMENSION A(n,n),B(N)
C**
C**     SOLVE EQUATIONS   [A] {X} = {B}
C**     [A]=N x N MATRIX
C**    {X}=UNKNOWN VECTOR, AFTER CALLING,{B}={X}
C**
        EPS=1.0D-10
        DO I=1,N
           do k=i+1,n
              IF (DABS(A(I,I)).GT.EPS) exit
              DO J=1,N
                 A(I,J)=A(I,J)+A(K,J)
              enddo
              B(I)=B(I)+B(K)
           enddo
           V=A(I,I)
C**
           DO J=1,N
              A(I,J)=A(I,J)/V
           enddo
           B(I)=B(I)/V
           DO M=1,N
              D=A(M,I)
              IF (M.NE.I) THEN
                 B(M)=B(M)-B(I)*D
                 DO J=1,N
                    A(M,J)=A(M,J)-A(I,J)*D
                 enddo
              ENDIF
           enddo
        enddo
C**
        end subroutine
```